
Данные для проектирования БИХ ЦФНЧ по АФНЧ (фильтр Чебышева)

Вариант 21

- верхняя граница затухания

  в полосе пропускания,  Δа [дБ]:          0,1

- нижняя граница затухания

  в полосе задерживания, а0[дБ]:            35

- граничная частота 

  полосы пропускания fг.п.[Гц]:               1080

- граничная частота 

   полосы задерживания fг.з.[Гц]:            2920

- частота дискретизации fд [Гц]:            8000

1. Расчет нормированных «цифровых» граничных частот.

    2. Расчет параметра преобразования γ.

3. Нахождение граничной «аналоговой» частоты ΩK полосы задерживания              АФ-прототипа.

4. Определение передаточной функции T(s).

Модуль коэффициента отражения (см. табл. с.241) по заданной величине ∆а       

     По номограмме определим вспомогательный параметр L по заданной величине a0 и определённой величине |p| а также порядок передаточной функции n 

общий вид передаточной функции будет иметь вид 

численные значения коэффициентов:

передаточная функция

    5. Определение передаточной функции ЦФНЧ H(z) (программа осуществляет билинейное преобразование передаточной функции АФНЧ в передаточную функцию ЦФНЧ (Л.М.Гольденберг, стр.111)).

20 REM      билинейное преобразование

40 DEFINT I, K-L

50 INPUT "введите коэффициент С для АФ "; C

60 INPUT "введите количество звеньев К "; K

70 DIM M(2, K), P(2, K), B1(2, K), A1(2, K)

80 PRINT "введите коэффициенты М(L,I), P(L,I) "

90 FOR I = 1 TO K

100 FOR L = 0 TO 2

110 PRINT "M("; L; ","; I; ")="; : INPUT M(L, I)

120 PRINT "P("; L; ","; I; ")="; : INPUT P(L, I)

130 NEXT L, I

140 INPUT "введите параметр GA "; GA

150 G1 = GA ^ 2: C1 = C

160 FOR I = 1 TO K

170 S1 = M(0, I) * G1: S2 = M(1, I) * GA: S3 = M(2, I)

180 IF P(0, I) = 0 THEN B1(0, I) = S2 + S3: B1(1, I) = S3 - S2: B1(2, I) = 0

190 IF P(O, I) <> 0 THEN B1(0, I) = S1 + S2 + S3: B1(1, I) = -2 * S1 + 2 * S3

200 IF P(0, I) <> 0 THEN B1(2, I) = S1 - S2 + S3

210 S1 = P(0, I) * G1: S2 = P(1, I) * GA: S3 = P(2, I)

220 IF P(0, I) = 0 THEN A1(0, I) = S2 + S3: A1(1, I) = S3 - S2: A1(2, I) = 0

230 IF P(0, I) <> 0 THEN A1(0, I) = S1 + S2 + S3: A1(1, I) = -2 * S1 + 2 * S3

240 IF P(0, I) <> 0 THEN A1(2, I) = S1 - S2 + S3

250 B0 = B1(0, I): A0 = A1(0, I)

260 C1 = C1 * B0 / A0

270 FOR L = 0 TO 2

280 B1(L, I) = B1(L, I) / B0

290 A1(L, I) = A1(L, I) / A0

300 NEXT L

310 NEXT I

320 PRINT "    коэффициенты БИХ-фильтра "

330 PRINT "     C="; C1: PRINT

340 FOR I = 1 TO K

350 PRINT "         звено        "; I

360 PRINT

370 PRINT "степень Z        числитель        знаменатель"                

380 PRINT

390 FOR L = 0 TO 2

400 L1 = -L

410 PRINT L1, B1(L, I), A1(L, I)

420 NEXT L

430 PRINT : PRINT

440 NEXT I

460 END              

Результат:                             
                                                          КОЭФ. БИХ-ФИЛЬТРА

                                                                 C=0,05907074
                                                                      ЗВЕНО        1

                                    СТЕПЕНЬ Z      ЧИСЛИТЕЛЬ     ЗНАМЕНАТЕЛЬ

                                               0                            1                             1

                                              -1                            1                    -0.3899166
                                              -2                            0                             0
                                                                      ЗВЕНО        2

                                    СТЕПЕНЬ Z      ЧИСЛИТЕЛЬ     ЗНАМЕНАТЕЛЬ

                                               0                            1                             1

                                              -1                            2                    -0,7334659
                                              -2                            1                      0,508058

    6. Построим АЧХ (программа осуществляет расчет АЧХ БИХ-фильтра по передаточной функции (Л.М. Гольденберг)).

10 REM     расчет АЧХ и затухания БИХ-фильтра

30 DEFINT I, K-M

40 INPUT "введите коэффициент С "; C

50 INPUT "введите количество звеньев К "; K

60 DIM B(2, K), A(2, K)

70 PRINT "введите коэффициенты B(L,I), A(L,I)"

80 FOR I = 1 TO K

90 FOR L = 0 TO 2

100 PRINT "B("; L; ","; I; ")="; : INPUT B(L, I)

110 PRINT "A("; L; ","; I; ")="; : INPUT A(L, I)

120 NEXT L, I

130 INPUT "введите количество расчетных точек М "; M

140 DW = .5 / (M - 1)

160 PRINT "частота          АЧХ          затухание   "

180 W = 0

190 FOR I1 = 1 TO M

200 W1 = 2 * 3.141592 * W

210 H = 1

220 FOR I = 1 TO K

230 S1 = 0: S2 = 0: S3 = 0: S4 = 0

240 FOR L = 0 TO 2

250 G1 = COS(L * W1): G2 = SIN(L * W1)

260 S1 = S1 + B(L, I) * G1: S2 = S2 + B(L, I) * G2

270 S3 = S3 + A(L, I) * G1: S4 = S4 + A(L, I) * G2

280 NEXT L

290 H = H * (S1 ^ 2 + S2 ^ 2) / (S3 ^ 2 + S4 ^ 2)

300 NEXT I

310 H = C * SQR(H)

320 IF H <> 0 THEN H1 = -20 * LOG(H) / LOG(10) ELSE H1 = 999

330 PRINT W; "   "; H; "    "; H1

350 W = W + DW

360 NEXT I1

380 PRINT "работа окончена"

390 END    

Результат: 

по таблице значений построим график АЧХ   

7. Построим импульсную и переходную характеристики с помощью программы.

10 REM    Расчет

20 REM   импульсной характеристики

30 REM   переходной характеристики

40 REM вых. сигнала при вх. сигнале sin(2*3.141592*N*W)

50 REM     К-каскадного БИХ-фильтра

70 DEFSTR Q

80 DEFINT I-L

90 INPUT "Введите коэффициент С"; C

100 INPUT "Введите количество звеньев"; K

110 DIM B(2, K), A(2, K), X(2, K)

120 PRINT "Введите коэффициенты B(L,I),A(L,I)"

130 PRINT "****вводите коэффициенты A(0,I)=0****"

140 FOR I = 1 TO K

150 FOR L = 0 TO 2

160 PRINT "B("; L; ","; I; ")="; : INPUT B(L, I)

170 PRINT "A("; L; ","; I; ")="; : INPUT A(L, I)

180 NEXT L, I

190 PRINT "   Введите "

200 PRINT "IR-при расчете импульсной характеристики"

210 PRINT "UR-при расчете переходной характеристики"

220 PRINT "SS-при расчете выходного сигнала "

230 INPUT Q

240 IF Q = "SS" THEN PRINT "Введите частоту синусоиды W"

250 IF Q = "SS" THEN INPUT W

260 INPUT "Введите количество расчётных точек M"; M

270 PRINT "   N   "; Q

280 FOR N = 0 TO M - 1

290 FOR I = 0 TO K

300 X(2, I) = X(1, I): X(1, I) = X(0, I)

310 NEXT I

320 IF Q = "IR" THEN GOSUB 490

330 IF Q = "UR" THEN GOSUB 510

340 IF Q = "SS" THEN GOSUB 520

350 X(0, 0) = XX

360 FOR I = 1 TO K

370 S = 0

380 FOR L = 0 TO 2

390 S = S + B(L, I) * X(L, I - 1) - A(L, I) * X(L, I)

400 NEXT L

410 X(0, I) = S

420 NEXT I

430 PRINT N; "   "; S

450 NEXT N

470 PRINT " Работа закончена "

480 END

490 IF N = 0 THEN XX = C ELSE XX = 0

500 RETURN

510 XX = C: RETURN

520 XX = C * SIN(2 * 3.141592 * N * W): RETURN

530 END

Импульсная характеристика

Результат: 

по таблице значений построим график

Переходная характеристика

Результат: 

по таблице значений построим график

